```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

We consider a partial differential system studied in J.-F. Pommaret, *Partial Differential Control Theory*, Kluwer Academic Publishers, Mathematics and Its Applications, 2001, p. 807, and defined by the following system matrix:

```
> A:=DefineOreAlgebra(diff=[d[1],x[1]],diff=[d[2],x[2]],diff=[d[3],x[3]],
> polynom=[x[1],x[2],x[3]]):
```

```
> R:=matrix(6,4,[0,-2*d[1],d[3]-2*d[2]-d[1],-1,0,d[3]-2*d[1],2*d[2]-3*d[1],1,d[3],
> -6*d[1],-2*d[2]-5*d[1],-1,0,d[2]-d[1],d[2]-d[1],0,d[2],-d[1],-d[2]-d[1],0,d[1],
> -d[1],-2*d[1],0]);
```

$$R := \begin{bmatrix} 0 & -2\,d_1 & d_3-2\,d_2-d_1 & -1 \\ 0 & d_3-2\,d_1 & 2\,d_2-3\,d_1 & 1 \\ d_3 & -6\,d_1 & -2\,d_2-5\,d_1 & -1 \\ 0 & d_2-d_1 & d_2-d_1 & 0 \\ d_2 & -d_1 & -d_2-d_1 & 0 \\ d_1 & -d_1 & -2\,d_1 & 0 \end{bmatrix}$$

Let us consider the ring $A = \mathbb{Q}[d_1, d_2, d_3]$ and $M = A^{1\times 4}/(A^{1\times 6}\,R)$ the A-module finitely presented by R. We denote by π the projection from $A^{1\times 4}$ to M. We can compute the A-module structure of the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A):
```

We find that E is generated by

```
> nops(Endo[1]);
```

$$11$$

elements which satisfy

```
> rowdim(Endo[2]);
```

$$18$$

A-linear relations.

Let us try to find idempotent elements of E defined by idempotent matrices $P \in \mathbb{Q}^{4\times 4}$ and $Q \in \mathbb{Q}^{6\times 6}$, namely, $e \in E$ satisfying $e^2 = e$, where $e(\pi(\lambda)) = \pi(\lambda\,P)$, for all $\lambda \in A^{1\times 4}$, and $R\,P = Q\,R$, $P^2 = P$, $Q^2 = Q$:

```
> Idem:=IdempotentsMatConstCoeff(S,Endo[1],A,0,alpha):
```

Let us consider the one defined by the following matrices $P_1 = Idem[1,1] \in A^{4\times 4}$, where we have set the arbitrary constants $c61$ and $c91$ appearing in P_1 to 0 and $Q_1 \in A^{6\times 6}$ satisfying $R\,P_1 = Q_1\,R$:

```
> P[1]:=subs(c61=0,c91=0,Idem[1,1]); Q[1]:=Factorize(Mult(R,P[1],A),R,A);
```

$$P_1 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ -1 & 1 & 2 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad Q_1 := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 2 \\ 2 & 1 & -1 & 0 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 & 6 \\ 0 & 0 & 0 & 1 & -1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

As we have $P_1^2 = P_1$ and $Q_1^2 = Q_1$, we know that the A-modules $\ker_A(.P_1)$, $\mathrm{im}_A(.P_1) = \ker_A(.(I_4 - P_1))$, $\ker_A(.Q_1)$ and $\mathrm{im}_A(.Q_1) = \ker_A(.(I_6 - Q_1))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of the corresponding modules. As the coefficients of P_1 and Q_1 belong to $\mathbb{Q}$, we can obtain them by means of linear algebraic techniques (e.g., using the *jordan* command of Maple) or using directly OreModules as it is explained below. We then form the matrices U_1 and V_1 such that $U_1\, P_1\, U_1^{-1}$ and $V_1\, Q_1\, V_1^{-1}$ are the Jordan normal forms of U_1 and V_1.

```
> U1:=SyzygyModule(P[1],A): U2:=SyzygyModule(evalm(1-P[1]),A):
> U[1]:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): V2:=SyzygyModule(evalm(1-Q[1]),A):
> V[1]:=stackmatrix(V1,V2);
```

$$U_1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & -1 & -2 & 0 \end{bmatrix} \quad V_1 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & -2 \\ 0 & 0 & 1 & 0 & 0 & -6 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 2 & 1 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

We obtain that the two unimodular matrices U_1 and V_1, i.e., $U_1 \in \mathrm{GL}_4(A)$ and $V_1 \in \mathrm{GL}_6(A)$, satisfy that the matrix $V_1\, R\, U_1^{-1}$ is block-diagonal:

```
> R_dec:=Mult(V[1],R,LeftInverse(U[1],A),A);
```

$$R\_dec := \begin{bmatrix} -2\,d_1 & 3\,d_1 + d_3 - 2\,d_2 & -1 & 0 \\ d_3 - 6\,d_1 & 7\,d_1 - 2\,d_2 & -1 & 0 \\ d_2 - d_1 & -d_2 + d_1 & 0 & 0 \\ 0 & 0 & 0 & -d_3 \\ 0 & 0 & 0 & -d_2 \\ 0 & 0 & 0 & d_1 \end{bmatrix}$$

We can also use the command *HeuristicDecomposition* to directly obtain the previous result:

```
> HeuristicDecomposition(R,P[1],A)[1];
```

$$\begin{bmatrix} -2\,d_1 & 3\,d_1 + d_3 - 2\,d_2 & -1 & 0 \\ d_3 - 6\,d_1 & 7\,d_1 - 2\,d_2 & -1 & 0 \\ d_2 - d_1 & -d_2 + d_1 & 0 & 0 \\ 0 & 0 & 0 & -d_3 \\ 0 & 0 & 0 & -d_2 \\ 0 & 0 & 0 & d_1 \end{bmatrix}$$

Let us now consider the first diagonal block S of R_{dec}:

```
> S:=submatrix(R_dec,1..3,1..3);
```

$$S := \begin{bmatrix} -2\,d_1 & 3\,d_1 + d_3 - 2\,d_2 & -1 \\ d_3 - 6\,d_1 & 7\,d_1 - 2\,d_2 & -1 \\ d_2 - d_1 & -d_2 + d_1 & 0 \end{bmatrix}$$

Let $N = A^{1\times 3}/(A^{1\times 3}\, S)$ be the A-module finitely presented by the matrix S. We can compute the A-module structure of the endomorphism ring $F = \mathrm{end}_A(N)$ of N:

```
> Endo1:=MorphismsConstCoeff(S,S,A):
```

Let us try to find idempotent elements of F defined by idempotent matrices $P_2 \in \mathbb{Q}^{3\times 3}$ and $Q_2 \in \mathbb{Q}^{3\times 3}$, namely, $S\,P_2 = Q_2\,S$, $P_2^2 = P_2$, $Q_2^2 = Q_2$:

```
> Idem1:=IdempotentsMatConstCoeff(S,Endo1[1],A,0,alpha):
```

We obtain non-trivial idempotent endomorphisms and we choose one of them defined by the following matrices $P_2 = Idem1[1,3] \in A^{3\times 3}$ and $Q_2 \in A^{3\times 3}$ satisfying $S\,P_2 = Q_2\,S$:

```
> P[2]:=Idem1[1,3]; Q[2]:=Factorize(Mult(S,P[2],A),S,A);
```

$$P_2 := \begin{bmatrix} 5/3 & -5/3 & 0 \\ 2/3 & -2/3 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad Q_2 := \begin{bmatrix} -2/3 & 2/3 & -4/3 \\ -5/3 & 5/3 & -4/3 \\ 0 & 0 & 1 \end{bmatrix}$$

As we have $P_2^2 = P_2$ and $Q_2^2 = Q_2$, we know that the A-modules $\ker_A(.P_2)$, $\mathrm{im}_A(.P_2) = \ker_A(.(I_3 - P_2))$, $\ker_A(.Q_2)$ and $\mathrm{im}_A(.Q_2) = \ker_A(.(I_3 - Q_2))$ are projective, and thus, free by the Quillen-Suslin theorem. Let us compute bases of the corresponding modules. As the coefficients of P_2 and Q_2 belong to $\mathbb{Q}$, we can obtain them by means of linear algebraic techniques (e.g., using the *jordan* command of Maple) or using directly OreModules as it is explained below. We then form the matrices U_2 and V_2 such that $U_2\,P_2\,U_2^{-1}$ and $V_2\,Q_2\,V_2^{-1}$ are the Jordan normal forms of U_2 and V_2.

```
> U1:=SyzygyModule(P[2],A): U2:=SyzygyModule(evalm(1-P[2]),A):
> U[2]:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[2],A): V2:=SyzygyModule(evalm(1-Q[2]),A):
> V[2]:=stackmatrix(V1,V2);
```

$$U_2 := \begin{bmatrix} 2 & -5 & 0 \\ 0 & 0 & 1 \\ 1 & -1 & 0 \end{bmatrix} \quad V_2 := \begin{bmatrix} 5 & -2 & 4 \\ 1 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

We obtain that the two unimodular matrices U_2 and V_2, i.e., $U_2 \in \mathrm{GL}_3(A)$ and $V_2 \in \mathrm{GL}_3(A)$, satisfy that the matrix $V_2\,S\,U_2^{-1}$ is block-diagonal:

```
> S_dec:=Mult(V[2],S,LeftInverse(U[2],A),A);
```

$$S\_dec := \begin{bmatrix} -d_1 + 2\,d_2 - d_3 & -3 & 0 \\ 0 & 0 & 4\,d_1 - d_3 \\ 0 & 0 & d_2 - d_1 \end{bmatrix}$$

Now, considering the following unimodular matrices X and Y defined by

```
> X:=Mult(diag(U[2],1),U[1],A);Y:=Mult(diag(V[2],1,1,1),V[1],A);
```

$$X := \begin{bmatrix} 2 & 0 & -5 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & -1 & 0 \\ 1 & -1 & -2 & 0 \end{bmatrix} \quad Y := \begin{bmatrix} 5 & 0 & -2 & 0 & 4 & -2 \\ 1 & 0 & -1 & 0 & 0 & 4 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 2 & 1 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

we can obtain the final decomposition of R in one step:

```
> R_final:=Mult(Y,R,LeftInverse(X,A),A);
```

$$R\_final := \begin{bmatrix} -d_1 + 2\,d_2 - d_3 & -3 & 0 & 0 \\ 0 & 0 & 4\,d_1 - d_3 & 0 \\ 0 & 0 & d_2 - d_1 & 0 \\ 0 & 0 & 0 & -d_3 \\ 0 & 0 & 0 & -d_2 \\ 0 & 0 & 0 & d_1 \end{bmatrix}$$

We obtain that the general solution $y = (y_1(x_1,x_2,x_3) \quad y_2(x_1,x_2,x_3) \quad y_3(x_1,x_2,x_3) \quad y_4(x_1,x_2,x_3))^T$ of the system $R\,y = 0$ is given by $X^{-1}\,z$ where $z = (z_1(x_1,x_2,x_3) \quad z_2(x_1,x_2,x_3) \quad z_3(x_1,x_2,x_3) \quad z_4(x_1,x_2,x_3))^T$ is the general solution of $R\_final\,z = 0$. The solutions of the latter system can be parametrized as follows:

```
>  z:=Parametrization(R_final,A);
```

$$z := \begin{bmatrix} 3\,\xi_1(x_1,x_2,x_3) \\ -\frac{\partial}{\partial x_1}\xi_1(x_1,x_2,x_3) + 2\,\frac{\partial}{\partial x_2}\xi_1(x_1,x_2,x_3) - \frac{\partial}{\partial x_3}\xi_1(x_1,x_2,x_3) \\ \_F1\,(x_3 + 1/4\,x_1 + 1/4\,x_2) \\ \_C1 \end{bmatrix}$$

For more details, see A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05. Then, we obtain the parametrization of the solutions of $R\,y = 0$ by applying the matrix X^{-1} to z:

```
>  y:=ApplyMatrix(LeftInverse(X,A),z,A);
```

$$y := \begin{bmatrix} -\xi_1(x_1,x_2,x_3) + 5/3\,\_F1\,(x_3 + 1/4\,x_1 + 1/4\,x_2) \\ \xi_1(x_1,x_2,x_3) + 1/3\,\_F1\,(x_3 + 1/4\,x_1 + 1/4\,x_2) - \_C1 \\ -\xi_1(x_1,x_2,x_3) + 2/3\,\_F1\,(x_3 + 1/4\,x_1 + 1/4\,x_2) \\ -\frac{\partial}{\partial x_1}\xi_1(x_1,x_2,x_3) + 2\,\frac{\partial}{\partial x_2}\xi_1(x_1,x_2,x_3) - \frac{\partial}{\partial x_3}\xi_1(x_1,x_2,x_3) \end{bmatrix}$$

Hence, the smooth solutions of $R\,y = 0$ are parametrized by an arbitrary function ξ_1 of three independent variables, an arbitrary function $\_F1$ of one independent variable and a constant $\_C1$. This result gives a simple proof of a result given in J.-F. Pommaret, *Partial Differential Control Theory*, Kluwer Academic Publishers, Mathematics and Its Applications, 2001, p. 807.

Finally, applying R to y, we can check that $R\,y = 0$:

```
>  ApplyMatrix(R,y,A);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

We refer the reader to A. Quadrat, D. Robertz, "Parametrizing all solutions of uncontrollable multidimensional linear systems", *Proceedings of* 16^{th} *IFAC World Congress*, Prague (Czech Republic), 04-08/07/05, for the proof that, conversely, every smooth solutions of $R\,y = 0$ has the previous form.